
Introduction

Now that we understand the formulation of demand in a typical
CGE model, we turn to the question of supply.

We’ll start with the basic cost minimization problem, and then turn
the problem of production in a competitive economy.

In the process, we’ll begin working with the CES function, which is
widely used in constructing various components of CGE models.

Session Outline

1 The cost minimization problem

2 Example using CES
3 Building the model in GAMS

Setting up the model
Calibration
Simulation and testing

John Gilbert Firms



Cost Minimization

Suppose that the firm uses inputs of labor (L) and capital (K ), for
which it must pay market prices w and r .

Its technology is described by the production function q = q(K , L).
This function represents the relationship between inputs and the
maximum output that can be produced, and is assumed to be
continuous and to exhibit diminishing returns to each factor and
CRTS.

The firm seeks to minimize its expenditure for a given level of
output, q̄.

Formal Problem

The problem can be written:

min L = rK + wL + λ[q̄ − q(K , L)]

The first order conditions are:

∂L /∂K = r − λ∂q/∂K = 0

∂L /∂L = w − λ∂q/∂L = 0

∂L /∂λ = q̄ − q(K , L) = 0

Interpretation

The solution to the minimization problem is the simultaneous
solution to these three equations for K , L and λ, expressed in terms
of w , r and q̄.

At an optimum, each factor price is equal to the value of the
marginal product of that factor.

Solving explicitly for the optimal input bundles yields the firm’s
factor demand functions.

John Gilbert Firms



Geometric Interpretation

Example - CES Technology

Suppose the firm’s technology can be described by the production
function q = γ[δKρ + (1− δ)Lρ]1/ρ, where ρ ≤ 1 and ρ 6= 0. The factor
demands will be:

K =
q̄δσr−σ

γ [(δr−ρ)σ + ((1− δ)w−ρ)σ]
1
ρ

L =
q̄(1− δ)σw−σ

γ [(δr−ρ)σ + ((1− δ)w−ρ)σ]
1
ρ

where σ = 1/(1− ρ) — the elasticity of substitution.

Set Notation

Let the set of factors be J = {K , L}. We can then let the factor demands
be represented by Fj , and the factor prices by rj . Using this notation, we
can write the factor demands compactly as:

Fj =
q̄δσj r

−σ
j

γ
[∑
∀k∈J

(
δk r
−ρ
k

)σ] 1
ρ

∀j ∈ J

Note how the summation term is defined over k rather than j .
Expenditure can be rewritten:

E =
∑
∀j∈J

rjFj

This is the form of the model that we will use in our GAMS program.

GAMS Program - Sets

Now let us consider exactly how the problem can be expressed in the
GAMS language.

Our first task is to create a set which will index the factors:

SET J Factors /K,L/ ;

ALIAS (J, JJ);

The keyword ALIAS defines another set called JJ that has exactly the
same elements as J.

GAMS Program - Parameters

Next, we need to define labels for all of the parameters and exogenous
variables in the model. We are also going to define labels for the initial
values of our endogenous variables:

PARAMETERS

GAMMA Shift parameter in production

DELTA(J) Share parameters in production

RHO Elasticity parameter in production

Q Output level

R(J) Factor prices

EO Initial expenditure

FO(J) Initial factor use levels;

GAMS Program - Variables

Our next task is to assign names for the variables:

VARIABLES

E Expenditure level

F(J) Factor use levels;

GAMS Program - Equations

We enter names for equations in the model in the same way that we
enter names for parameters and variables:

EQUATIONS

EXPENDITURE Expenditure function

FDEMAND(J) Factor demand functions;

Then we define the structure of the equations in terms of the variables
and parameters:

EXPENDITURE..E=E=SUM(J, R(J)*F(J));

FDEMAND(J)..F(J)=E=(Q/GAMMA)*DELTA(J)**(1/(1-RHO))*

R(J)**(-1/(1-RHO))/SUM(JJ, (DELTA(JJ)*

R(JJ)**(-RHO))**(1/(1-RHO)))**(1/RHO);

These are the GAMS equivalents of the equations we derived.

GAMS Program - Calibration

With the labels created, we can calibrate:

R(J)=1;

Q=100;

FO(J)=50;

EO=SUM(J, R(J)*FO(J));

RHO=0.1;

DELTA(J)=(R(J)/FO(J)**(RHO-1))/(SUM(JJ, R(JJ)/FO(JJ)**

(RHO-1)));

GAMMA=Q/(SUM(J, DELTA(J)*FO(J)**RHO))**(1/RHO);

GAMS Program - Initial Levels and Bounds

Next we set the starting values:

E.L=EO;

F.L(J)=FO(J);

and assign lower bounds:

F.LO(J)=0;

GAMS Program - Solve

The last stage is to tell GAMS which of our equations constitute the
model, in this case all of them, and then run a test solution:

MODEL PRODUCTION /ALL/;

SOLVE PRODUCTION USING NLP MINIMIZING E;

Exercises

We have now built our second simple numerical simulation model.
Once we are satisfied that the model is correctly replicating the
solution, we can examine the effects of changes in the economic
situation by altering the values of parameters and/or exogenous
variables and executing another SOLVE command.

Some things to try:

Changing factor prices and/or expenditure in various combinations.
Altering the model parameters.
Experimenting with different functional forms.

Further Reading

Most of this material is covered in intermediate microeconomics
texts such as Varian (2009).

Dixit (1990) is a good introduction to optimization.

This GAMS example is developed more fully in Gilbert and Tower
(2012), chapter 4.

